5. PRAKTIKA

ILARA-TEORIA

Hainbat antolakuntzan-gertaeran bi izakik parte hartzen dute. Lehenengo izakiak, *zerbitzariak* bezala ezagutzen direnak, sarritan diskretuak izaten dira (pertsonak, makinak, autobideko ordainlekuetako kabinak, …). Bigarren izakiak ordea, *bezeroak* bezala ezagutzen dira eta zerbitzariak erabiltzen dituztenak izaten dira. Sistema mota hauek *ilara-teoria* bezala ezagutzen dira.

Ilara teoriak A.K. Erlang (1909) lanean du jatorria, lan honetan telefono-deien metaketa aztertzen da. Ilara eredurik errazena irudian ikus daitekeena da, bertan bezeroak *sarrerara* iturri finitu edo infinitu batetik heltzen dira eta *zerbitzu kanalean* zerbitzatuak izan arte ilaran itxaroten dute. Bukatzeko zerbitzatuak izan ondoren, erabiltzaileek aurrera zerbitzuaren puntu berri batera edo irteerara joaten dira.



Infinitu ilara mota daudela dirudi. Guztiak sinplifikatzeko, eredu posibleak (*Kendall notazioa* jarraituz) ondorengo sailkapena erabiliz adierazten dira:

$$X/Y/s/K/A/B$$

X: bezeroen helduearen banaketa.

Y: zerbitzu-denboraren banaketa.

s*:* paraleloan lan egiten duten zerbitzari kopurua.

K*:* ilararen ahalmena (ilaran eta zerbitzuan egon ahal den bezero kopurua); balio lehenetsia ∞ da.

A*:* populazioaren tamaina; balio lehenetsia ∞ da.

***B**:* zerbitzu diziplina; balio lehenetsia FIFO da.

s*, *K eta ***A*** balioak zenbakizko balio egokiengatik ordezkatzen dira eta *X* eta *Y* banaketentzat erabilienak diren laburdurak ondorengoak dira:

***M**:* ondoz-ondoko heldueren artean igarotzen den denboraren banaketa esponentziala da (Poisson-en helduerak) edo zerbitzu-denboraren banaketa esponentziala da.

***G**:* Banaketa orokorra.

***D**:* banaketa determinista (endekatua).

***E**$_k$:* *k* bigarren parametroa duen Erlang-en banaketa

B zerbitzuaren diziplina hurrengo motakoa izaten da:

***FIFO* (*first in-first out*)**: lehena heltzen dena zerbitzatua izaten edo irteten lehena da.

***LIFO* (*last in-first out*):** azkena heltzen dena zerbitzatua izaten lehena da.

***SIRO* (*service in random order*):** zerbitzatua izateko ordena zorizkoa da.

GD (*general service discipline):* zerbitzu orokorreko diziplina.

RR *(round robin):* bezeroa denbora finko batean zehar zerbitzatua izaten da, bere eskariak betetzen ez badira berriro ere ilarara bueltatuz (ordenagailuen sareak).

Adibidez ***M/D/1/∞/∞/FIFO*** notazioak Poisson-en helduerak, zerbitzu denbora determinista (konstantea), zerbitzari bakarra, bezero-kopuru mugagabea, populazio infinitua eta *FIFO* diziplina dituen ilara adierazten du. Hirugarren baliotik aurrera agertzen diren balioak lehenetsiak direnez, aurreko adierazpena *M/D/1* bezala laburbiltzen da.

Gainera, *egoera egonkorra* lortu ondoren ilara baten funtzionamendua edo jarduera neurtzeko hainbat oinarrizko ezaugarri daude. Garrantzitsuenak ondorengoak dira:

***N**:* sisteman (ilaran eta zerbitzuan) dagoen bezero kopurua adierazten duen zorizko aldagaia.

***N**$_q$:* ilaran dagoen bezero kopurua adierazten duen zorizko aldagaia.

***T**:* bezero batek sisteman igarotako denbora adierazten duen zorizko aldagaia.

T_q: bezero batek ilaran igarotako denbora adierazten duen zorizko aldagaia.

P_n: sisteman zehatz-mehatz *n* bezero egoteko probabilitatea

L: sisteman itxaroten den bezero kopurua

L_q: ilaran egotea (zerbitzatuak izateko zain) itxaroten den bezero kopurua

W: sisteman bezero batek igarotzen duen batezbesteko itxarote-denbora.

W_q: ilaran bezero batek igarotzen duen batezbesteko itxarote-denbora.

Bestalde, sisteman dagoen zirkulazioa hurrengoek zehazten dute:

λ_n : bezeroen heldueren batezbesteko tasa sisteman *n* bezero daudenean.

μ_n : zerbitzuaren batezbesteko tasa sisteman *n* bezero daudenean.

λ : bezeroen heldueren batezbesteko tasa, edozein *n* baliotarako konstantea denean.

μ : bezeroen zerbitzuaren batezbesteko tasa, edozein *n* baliotarako konstantea denean.

$\rho = \frac{\lambda}{\mu}$: erabilera-faktorea edo trafiko-intentsitatea.

Jarraian ohikoenak diren ilara ereduak azalduko dira. Arreta beraien oinarrizko ezaugarrien formulazioan jarriko da eta adibide praktikoak Excel kalkulu orria erabiliz ebatziko dira.

M/M/1 eredua

Ilara honetan elkarren segidan heltzen diren bezeroen heldueren arteko denborak λ parametrodun banaketa esponentziala jarraitzen du (zorizko helduerak, Poisson banaketa jarraitzen dutenak), zerbitzatuak izateko denborak μ parametrodun banaketa esponentziala jarraitzen du eta sisteman egon daitezkeen bezero kopurua mugagabea da. Ilara egonkorra izateko, hau da, ilara infinitua sor ez dadin, $\rho < 1$ izan behar da.

Bere oinarrizko ezaugarri nagusiak ondorengoak dira:

$$P_0 = 1 - \rho$$
$$P_n = \rho^n P_0$$

$$L_q = \frac{\rho^2}{1-\rho}$$

$$L = \rho + L_q = \frac{\rho}{1-\rho}$$

$$W_q = \frac{\rho}{\mu(1-\rho)}$$
$$W = W_q + \frac{1}{\mu} = \frac{1}{\mu(1-\rho)}$$

Sisteman gutxienez bezero bat egoteko probabilitatea (bezero batek itxaron behar izateko probabilitatea) hurrengoa da

$$P(N>0) = 1 - P_0 = \rho$$

Sisteman *n* bezero baino gehiago egoteko probabilitatea ondokoa da

$$P(N>n) = \rho^{n+1}$$

Zerbitzaria egonean egoteko probabilitatea edo bezeroek itxaron behar ez izateko probabilitatea ondorengoa da

$$P(N=0) = P_0 = 1 - \rho$$

Bezeroak sisteman igarotzen duen denboren, hau da *T*-ren, banaketa hurrengoa da

$$P(T \le t) = 1 - e^{-\mu t(1-\rho)}$$

Ilaran itxaron behar den denbora neurtzen duen zorizko aldagaiaren banaketa, hau da T_q-ren banaketa funtzioa ondokoa da

$$P(T_q \le t) = 1 - \rho e^{-\mu t(1-\rho)}$$

M/M/1 ereduaren adibidea

Etxetresna elektrikoak saltzen dituen denda batera bezeroak Poisson-en banaketa jarraituz heltzen dira, 12 minutuko helduera-tasa 1 izanik (orduro 5 bezero). Bestalde, bezero bakoitza zerbitzatzeko batez beste 5 minutu behar izaten dira (orduro 12 bezeroko tasa izanik eta zerbitzatuak izateko behar izaten den denborak banaketa esponentziala jarraituz). Ondorengoak zehaztu nahi dira:

Bezero berri batek, zerbitzatua izan aurretik 10 minutu (1/6 ordu) baino gehiago itxaron behar izateko probabilitatea.

Ilaran 3 pertsona (sisteman 4 pertsona) baino gehiago egoteko probabilitatea.

Bezero batek bere kontsulta osatzeko 10 minutu (1/6 ordu) baino gehiago itxaron behar izateko probabilitatea.

N zorizko aldagaiaren, hau da, sisteman dagoen bezero-kopurua adierazten duen aldagaiaren probabilitate-funtzioa.

Soluzioa:

	A	B	C	G	H
1	M/M/1 ilara infinitua- Populazio infinitua				
2					
3	Helduera-tasa	5			
4	Zerbitzu-tasa	12			
5	Denbora-unitatea	*ordua*			
6					
7	Trafiko-intentsitatea	=B3/B4		n	P(n)
8	isteman bezerorik ez egoteko probabilitatea P(0)	=1-B7		0	=B8
9	Sisteman bezeroren bat egoteko probabilitatea	=B7		1	=(B7^G9)*B8
10	Lq	=B7^2/(1-B7)		2	=(B7^G10)*B8
11	L	=B7/(1-B7)		3	=(B7^G11)*B8
12	Wq	=B7/(B4*(1-B7))		4	=(B7^G12)*B8
13	W	=1/(B4*(1-B7))		5	=(B7^G13)*B8
14				6	=(B7^G14)*B8
15	Bezeroak ilaran igarotzen duen denbora t	=1/6		7	=(B7^G15)*B8
16	P(Tq>t)	=B7*EXP(-B4*(1-B7)*B15)		8	=(B7^G16)*B8
17	N bezero-kopurua	4		9	=(B7^G17)*B8
18	P(N>n)	=B7^(B17+1)		10	=(B7^G18)*B8
19	Bezeroak sisteman igarotzen duen denbora t	=1/6		11	=(B7^G19)*B8
20	P(T<=t)	=1-EXP(-B19*(1-B7)*B4)		12	=(B7^G20)*B8
21	P(T>t)	=1-B20		13	=(B7^G21)*B8
22				14	=(B7^G22)*B8

	A	B	C	G	H
1	M/M/1 ilara infinitua- Populazio infinitua				
2					
3	helduera-tasa	5			
4	Zerbitzu-tasa	12			
5	Denbora-unitatea	*ordua*			
6					
7	Trafiko-intentsitatea	0,416667		n	P(n)
8	Sisteman bezerorik ez egoteko probabilitatea P(0)	0,583333		0	0,583333333
9	Sisteman bezeroren bat egoteko probabilitatea	0,416667		1	0,243055556
10	Lq	0,297619		2	0,101273148
11	L	0,714286		3	0,042197145
12	Wq	0,059524		4	0,017582144
13	W	0,142857		5	0,007325893
14				6	0,003052456
15	Bezeroak ilaran igarotzen duen denbora t	0,16667		7	0,001271856
16	P(Tq>t)	0,129751		8	0,00052994
17	N bezero-kopurua	4		9	0,000220808
18	P(N>n)	0,012559		10	9,20035E-05
19	Bezeroak sisteman igarotzen duen denbora t	0,16667		11	3,83348E-05
20	P(T<=t)	0,688597		12	1,59728E-05
21	P(T>t)	0,311403		13	6,65535E-06
22				14	2,77306E-06

Irudian emaitzak eta sistemaren ezaugarriak agertzen dira. Datu hauek erabiliz, planteatutako galderen erantzunak ondorengoak dira:

Bezero berri batek saltzaile batek zerbitzatzeko 10 minutu baino gehiago itxaron behar izateko probabilitatea oso baxua da (0,12975).

Ilaran 3 pertsona (sisteman 4 pertsona) baino gehiago egoteko probabilitatea ia nulua da (0,01256).

Bezero batek bere kontsulta osatzeko 10 minutu baino gehiago itxaron behar izateko probabilitatea, hau da, dendara sartzen denetik irteten den arte igarotako denbora 10 minutu baino gehiago izateko probabilitatea 0,3114 da.

Sisteman dagoen bezero-kopurua adierazten duen zorizko aldagaiaren probabilitate-funtzioa hurrengo grafikoan agertzen dena da:



M/M/s eredua

Eredu honetan *s* zerbitzari kontsideratzen dira. Ilara egonkorra izateko baldintza $\rho^* < 1$ da, $\rho^* = \dfrac{\lambda}{s\mu}$ *trafikoaren intentsitate globala* bezala ezagutzen da.

Ilara honen funtzionamendurako ezaugarri nagusiak ondorengoak dira

$$P_0 = \frac{1}{\dfrac{\rho^s}{s!(1-\rho^*)} + \displaystyle\sum_{k=0}^{s-1} \frac{\rho^k}{k!}}$$

$$P_n = \begin{cases} \dfrac{\rho}{n} P_{n-1} = P_0 \dfrac{\rho^n}{n!} & si \;\; 1 \le n \le s \\ \left(\rho^*\right)^{n-s} P_s & si \;\; n > s \end{cases}$$

$$L = \rho + \frac{\rho^*}{(1-\rho^*)^2} P_s$$

$$L_q = L - \rho$$

$$W_q = \frac{P_s}{s\mu(1-\rho^*)^2}$$

$$W = W_q + \frac{1}{\mu}$$

Egonean dauden batezbesteko zerbitzari kopurua $s - \rho$ izanik.

Ilaran itxaron behar izateko probabilitatea, hau da, sisteman *s* bezero edo bezero gehiago egoteko probabilitatea (*Erlang-en formula*)

$$P(N \geq s) = \frac{\rho^s}{s!(1-\rho^*)} P_0 \qquad \text{da.}$$

T_q aldagaiaren, hau da, ilaran itxaron behar den denbora neurtzen duen zorizko aldagaiaren banaketa funtzioa ondokoa da

$$P(T_q \leq t) = 1 - \frac{P_s e^{-\mu t(s-\rho)}}{1-\rho *}$$

M/M/s ereduaren adibidea

Autobide bateko ordainlekuak 5 zerbitzari automatiko ditu, zerbitzatuak izateko denborak minutu bateko batezbestekoa (orduko 60 bezeroko tasa) duen banaketa esponentziala jarraituz. Ibilgailuen helduerek orduko 299 batezbestekoa duen Poisson-en banaketa jarraitzen badute, bi zerbitzari gehiago jarri beharko al dira itxaron behar izaten den batezbesteko denbora murrizteko eta autobidean ematen den zerbitzuari buruz dagoen iritzia aldatzeko?

Hurrengo irudietan autobideak bost zerbitzari dituenean Excel kalkulu

	A	B	C	D	E	F	G
1							M/M/s Ilara infinitua - Populazio infinitua
2							
3	Helduera-tasa	299					
4	Zerbitzu-tasa	60					
5	Zerbitzari kopurua	5					
6	Denbora-unitatea	*Ordua*			=B3/B4		
7							
8	Trafiko-intentsitatea	=B3/(B4*B5)					P(n)
9	P(0)	=G9				0	=1/((E6^5/(FACT(B5)*(1-B8)))+1+E6+(E6^2)/2+(E6^3)/6+(E6^4)/24)
10	L	=E6+(B8*G14/(1-B8)^2)				1	=G9*E6
11	Zerbitzu-automatikoak erabiltzeko itxaroten den kotxe kopurua Lq	**=B10-E6**				2	=G9*E6^2/FACT(2)
12	Wq	=G14/(B5*B4*(1-B8)^2)				3	=G9*E6^3/FACT(3)
13	Gidari batek ordainlekua pasatzeko behar duen batezbesteko denbora W	**=B12+1/B4**				4	=G9*E6^4/FACT(4)
14						5	=G9*E6^5/FACT(5)

orria erabiliz lortzen diren emaitzak ikus daitezke.

	A	B	C	D	E	F	G
1		**M/M/s Ilara infinitua - Populazio infinitua**					
2							
3	**Helduera-tasa**	**299**					
4	**Zerbitzu-tasa**	**60**					
5	**Zerbitzari kopurua**	**5**					
6	**Denbora-unitatea**	*Ordua*			4,98333		
7							
8	**Trafiko-intentsitatea**	0,996666667					P(n)
9	**P(0)**	0,000129068				0	0,000129068
10	**L**	301,4856328				1	0,000643187
11	**Zerbitzu-automatikoak erabiltzeko itxaroten den kotxe kopurua Lq**	**296,5022995**				2	0,001602607
12	**Wq**	0,991646487				3	0,002662108
13	**Gidari batek ordainlekua pasatzeko behar duen batezbesteko denbora W**	**1,008313153**				4	0,003316543
14						5	0,003305488

Irudian ikusten den bezala, gidari batek ordainlekua pasatzeko batez beste ordu bat baino gehiago (1,0083 h) behar du, denbora gehiena (0,9916 h) zerbitzatua izateko itxaroten emango duelarik. Ondorioz, ordainlekuan auto-ilara handiak egongo dira.

Bi zerbitzari gehiago jarriz gero, ondorengo emaitzak lortzen dira:

1		**M/M/s Ilara infinitua - Populazio infinitua**					
2							
3	**Helduera-tasa**	**299**					
4	**Zerbitzu-tasa**	**60**					
5	**Zerbitzari kopurua**	**7**					
6	**Denbora-unitatea**	*Ordua*		4,98			
7							
8	**Trafiko-intentsitatea**	0,7119048				P(n)	
9	**P(0)**	0,0060914			0	0,0060914	
10	**L**	5,7745156			1	0,0303555	
11	**Zerbitzu-automatikoak erabiltzeko itxaroten den kotxe kopurua Lq**	**0,791182**			2	0,0756358	
12	**Wq**	0,0026461			3	0,1256395	
13	**Gidari batek ordainlekua pasatzeko behar duen batezbesteko denbora W**	**0,019313**			4	0,1565259	
14					5	0,1560042	
15					6	0,1295701	
16					7	0,0922416	

Irudian ikus daitekeen moduan, gidari batek ordainlekua pasatzeko behar duen batezbesteko denbora 0,019 ordukoa da, minutu bat baino pixka bat handiagoa. Itxaronaldi hau aurreko kasuan lortutakoa baino askoz txikiagoa da.

Hurrengo irudietan ikusten den moduan, kabina bi jarri ordez kabina bakarra jartzeak ere auto-ilarak ekidingo ditu.

	A	B	C	D	E	F	G
1		**M/M/s Ilara infinitua - Populazio infinitua**					
2							
3	**Helduera-tasa**	**299**					
4	**Zerbitzu-tasa**	**60**					
5	**Zerbitzari kopurua**	**6**					
6	**Denbora-unitatea**	*Ordua*			4,98		
7							
8	**Trafiko-intentsitatea**	0,830555556					P(n)
9	**P(0)**	0,004632502				0	0,004633
10	**L**	7,833805134				1	0,023085
11	**Zerbitzu-automatikoak erabiltzeko itxaroten den kotxe kopurua Lq**	**2,850471801**				2	0,057521
12	**Wq**	0,009533351				3	0,095549
13	**Gidari batek ordainlekua pasatzeko behar duen batezbesteko denbora W**	**0,026200017**				4	0,119038
14						5	0,118641
15						6	0,098538

Bukatzeko, 5 zerbitzari automatiko badaude, zerbitzatuak izateko denborak minutu bateko batezbestekoa duen banaketa esponentziala jarraitzen badute eta helduera tasa 300 ibilgailuko bada, trafikoaren intentsitate globala bat izango dela eta ondorioz sistema ezegonkorra izango dela ohartarazi nahi da, ilarak luzera infinitua izanik.

M/G/1 eredua

Aurreko bi ereduak Poisson-en ilarak dira, izan ere heldueren arteko denborak eta zerbitzatuak izateko denborak banaketa esponentziala jarraitzen dute. Azken suposizio hau ez da egoera praktikoetara ondo doitzen, orokorrean errealitatean zerbitzatuak izateko denborak ez baitu banaketa esponentziala jarraitzen, batez ere bezeroen eskakizunak antzekoak direnean. Zoritxarrez, nahiz eta eredu batzuentzat emaitza erabilgarri batzuk lortu izan diren, banaketa ez-esponentzialak dituzten ilaren azterketa analitikoa askoz zailagoa da.

M/G/1 ereduan helduerek λ parametrodun Poisson-en banaketa (heldueren arteko denborak banaketa esponentziala) jarraitzen dutela, zerbitzatuak izateko denborak hautazko banaketa duela eta zerbitzari bakarra dagoela suposatzen da. $1/\mu$ zerbitzu-denboraren batezbestekoa eta σ^2 bariantza ezagunak badira, $\rho = \dfrac{\lambda}{\mu} < 1$ egonkortasun baldintza lortzen denean ondorengo emaitzak ezar daitezke:

$$P_0 = 1 - \rho$$

$$L_q = \frac{\lambda^2\sigma^2 + \rho^2}{2(1-\rho)}$$

Azken formula hau **Pollaczek-Kyntchine-en formula** da eta M/G/1 ereduak praktikan duen ugaritasuna dela eta ilara teoriko emaitzarik garrantzitsuenetariko bat da. Gainera,

$$L = \rho + L_q$$

$$W_q = \frac{\lambda(\sigma^2 + \frac{1}{\mu^2})}{2(1-\rho)}$$

$$W = \frac{1}{\mu} + W_q$$

M/G/1 ereduaren ilara

Mediku batengana doazen gaixoak anbulatoriora Poisson-en banaketa jarraituz heltzen dira, orduko helduera-tasa 3 izanik. Medikuaren kontsulta-denboraren banaketak gaixo bakoitzeko 12 minutuko batezbestekoa eta 24 minutuko desbideratze tipikoa ditu. Zein izango da gaixo batek zain egon behar izango duen denbora? Zein izango da itxarongelan dagoen batezbesteko gaixo-kopurua?

Irudietan kalkulu orrian egin beharreko programazioa, programatu ondoren lortutako emaitzak eta sistemaren ezaugarriak agertzen dira.

	B	C	D	E	F	
1						
2	M/G/1 Ilara infinitua - Populazio infinitua					
3						
4	Helduera-tasa	3	*bezero orduk*			
5	zerbitzu-denboraren batezbestekoa	0,2	*orduak*	Zerbitzu-tasa	=1/C5	*bezero /ordu*
6	zerbitzu-denboraren desbidazio tipikoa	0,4	*orduak*			
7	denbora-unitatea	*ordua*				
8						
9	Trafiko-intentsitatea	=C4/F5				
10	Itxarongelan dagoen batezbesteko gaixo-kopurua Lq	=(C4^2*C6^2+C9^2)/(2	*bezeroak*			
11	Sisteman dagoen batezbesteko gaixo-kopurua L	=C9+C10	*bezeroak*			
12	Iitxarongelan gaixo batek igarotzen duen batezbesteko denbora Wq	=C10/C4	*orduak*			
13	Sisteman gaixo batek igarotzen duen batezbesteko denbora W	=C12+1/F5	*orduak*			
14						

	A / B	C	D	E	F	G
2	M/G/1 Ilara infinitua - Populazio infinitua					
3						
4	Helduera-tasa	3	*bezero orduko*			
5	zerbitzu-denboraren batezbestekoa	0,2	*orduak*	Zerbitzu-tasa	5	*bezero /ordu*
6	zerbitzu-denboraren desbidazio tipikoa	0,4	*orduak*			
7	denbora-unitatea	*ordua*				
8						
9	Trafiko-intentsitatea	0,6				
10	Itxarongelan dagoen batezbesteko gaixo-kopurua Lq	2,25	*bezeroak*			
11	Sisteman dagoen batezbesteko gaixo-kopurua L	2,85	*bezeroak*			
12	Iitxarongelan gaixo batek igarotzen duen batezbesteko denbora Wq	0,75	*orduak*			
13	Sisteman gaixo batek igarotzen duen batezbesteko denbora W	0,95	*orduak*			
14						

Kontsultara sartzeko gaixo batek itxaron behar duen batezbeteko denbora 0,75 ordukoa (45 minutukoa) da, kalera irteteko ordea 0,95 ordu (57 minutu) behar izango ditu. Hau da, medikuarekin 12 minutu egoteko gaixoak 45 minutu itxaron behar ditu. Bestalde, gutxienez itxarongelan batez beste bi gaixo egongo dira.

L_q-ren eta osasun-gunera heltzen diren gaixoen helduera-tasaren arteko erlazioa aztertzen bada, ondorengo irudian ikusten den bezala, hazkunde esponentziala dagoela ikus daiteke.



	A	B	C	D	E	F	G	H	I	J	K	L	M
1	***Lq-ren eta helduera-tasaren arteko erlazioa***												
2													
3	**Helduera-tasa**	3											
4	**Zerbitzu-denboraren batezbestekoa**	0,2											
5	**Zerbitzu-denboraren desbideratze tipikoa**	0,4											
6	**Intentsitate-trafikoa**	0,6											
7	**Lq**	2,25											
8													
9		LAMBDA=	**3**	**0**	**0,5**	**1**	**1,5**	**2**	**3**	**3**	**3,5**	**4**	**4,5**
10		Lq=	2,25	0	0,03	0,1	0,3	1	1	2,3	4,08	8	20

M/D/1 eredua

Zerbitzariak behin eta berriz ia zerbitzu berdina ematen duenean, hau da, zerbitzua ohiko zeregin bera egiten oinarritzen denean, zerbitzu-denbora $1/\mu$ konstantea da (zerbitzu-denboraren banaketa endekatua da). Izan ere, zerbitzua emateko beharrezkoa den denbora gutxi aldatzen baita.

Bestalde M/D/1 ilaran helduerak λ parametrodun Poisson-en banaketa jarraitzen dutela eta zerbitzari bakarra dagoela suposatzen da.

Kasu honetan $\rho < 1$ betetzen bada, ondorengoa lortzen da:

$$L_q = \frac{\rho^2}{2(1-\rho)} \qquad W_q = \frac{\rho}{2\mu(1-\rho)}$$

$$L = \rho + L_q \qquad W = \frac{1}{\mu} + W_q$$

M/D/1 ereduaren adibidea

Eskuz egindako egurrezko ehungailu batean hari bat edo hari talde bat igotzen dituen mekanismoa oso ondo doituta egon behar da. Mekanismo hau matxuratzen bada langileak zehazki 12 minutu behar ditu berriro ere prest jarri eta ehuntzen jarraitzeko. Orduro batezbeste 3 ehungailu desdoitzen badira (matxurek Poisson-en *prozesua jarraituz), zehaztu desdoitutako ehungailu bat konpontzen hasteko itxaron behar den batezbesteko denbora.*

Hurrengo irudietan ilararen funtzionamenduaren parametro nagusiak azaltzen dira, hauen artean, W_q eskatutako batezbesteko denbora eta kalkulu orrian egin beharreko programazioa.

	A	B	C	D	E	F
1	**M/D/1 Ilara infinitua -Populazio infinitua**					
2						
3	**Helduera-tasa**	3	*ehungailu orduko*			
4	**Zerbitzuaren iraupena**	0,2	*ordu*	Zerbitzu-tasa	**=1/B4**	*ehungailu orduko*
5	**Denbora-unitatea**	*ordua*				
6						
7	**Intentsitate-trafikoa**	**=B3/E4**				
8	**Lq**	**=B7^2/(2*(1-B7))**	*ehungailu*			
9	**L**	**=B7+B8**	*ehungailu*			
10	**Desdoitutako ehungailu bat konpontzen hasteko itxaron behar den batezbesteko denbora Wq**	**=B8/B3**	*ehungailu*			
11	**W**	**=B10+1/E4**	*ehungailu*			

	A	B	C	D	E	F
1	**M/D/1 Ilara infinitua -Populazio infinitua**					
2						
3	**Helduera-tasa**	**3**	*ehungailu orduko*			
4	**Zerbitzuaren iraupena**	**0,2**	*ordu*	Zerbitzu-tasa	**5**	*ehungailu orduko*
5	**Denbora-unitatea**	*ordua*				
6						
7	**Intentsitate-trafikoa**	**0,6**				
8	**Lq**	**0,45**	*ehungailu*			
9	**L**	**1,05**	*ehungailu*			
10	**Desdoitutako ehungailu bat konpontzen hasteko itxaron behar den batezbesteko denbora Wq**	**0,15**	*ehungailu*			
11	**W**	**0,35**	*ehungailu*			

Ehungailu bat konpontzen hasteko itxaron behar den batezbesteko denbora 0,15 ordukoa da (9 minutukoa), eta era egokian berriro ere lan egiteko zehazki 0,35 ordu (21 minutu) behar izaten dira. Laburbilduz, ehungailua konpontzen hasteko 9 minutu itxaron behar izaten dira eta langilea hura konpontzen 12 minutu egoten da. Ondorioz, konpontzen egoten den ehungailu kopurua gutxi gorabehera bat izaten da (1,05 ehungailu).